# Maintainer: Christoph Reiter <reiter.christoph@gmail.com>

_realname=tzdata
pkgbase=mingw-w64-${_realname}
pkgname=("${MINGW_PACKAGE_PREFIX}-${_realname}")
pkgver=2023c
pkgrel=1
pkgdesc="Sources for time zone and daylight saving time data (mingw-w64)"
arch=('any')
mingw_arch=('mingw32' 'mingw64' 'ucrt64' 'clang64' 'clang32' 'clangarm64')
url="https://www.iana.org/time-zones"
license=('custom: public domain')
options=('!emptydirs')
makedepends=('tzcode')
source=(https://www.iana.org/time-zones/repository/releases/${_realname}${pkgver}.tar.gz{,.asc})
sha512sums=('608bd286ebcbd0004cfdc1da183273f08aff61f90c8867661154453d77a05d421e4c46ad6d066a1fe2e87d5c82ec0f1c0224667a3b35f3180a3eb7f6ff84cbf5'
            'SKIP')
validpgpkeys=('7E3792A9D8ACF7D633BC1588ED97E90E62AA7E34') # Paul Eggert <eggert@cs.ucla.edu>

_timezones=('africa' 'antarctica' 'asia' 'australasia'
           'europe' 'northamerica' 'southamerica'
           'etcetera' 'backward' 'factory')

package() {
  cd "${srcdir}"

  # install tzdata stuff
  zic -b fat -d "${pkgdir}${MINGW_PREFIX}"/share/zoneinfo ${_timezones[@]}
  zic -b fat -d "${pkgdir}${MINGW_PREFIX}"/share/zoneinfo/posix ${_timezones[@]}
  zic -b fat -d "${pkgdir}${MINGW_PREFIX}"/share/zoneinfo/right -L leapseconds ${_timezones[@]}
  # This creates the posixrules file. We use New York because POSIX requires the daylight savings time rules to be in accordance with US rules.
  zic -b fat -d "${pkgdir}${MINGW_PREFIX}"/share/zoneinfo -p America/New_York
  install -m444 -t "${pkgdir}${MINGW_PREFIX}"/share/zoneinfo iso3166.tab leap-seconds.list zone1970.tab zone.tab # zone.tab is depricated and will go soon

  # install license
  install -Dm644 LICENSE "${pkgdir}${MINGW_PREFIX}"/share/licenses/${_realname}/LICENSE
}
